
Erfahrungen mit NanoESP oder Pretzel Board 

Teil 6: NanoESP & Funkmodul 433 MHz

433 MHz Sender

Hinweise:
( ">", meint Menü auswählen)

Die Idee:
Auf dem NanoESP (evtl. Arduino UNO mit ESP8266) läuft ein Webserver.
Die Webseite stellt Schaltflächen bereit, mit denen mit dem NanoESP verbundene Hardware 
gesteuert werden soll.
In unserem Projekt soll ein 433 MHz-Sender über Funksignale Funksteckdosen steuern.

433 MHz Sende & Empfänger
Funksteckdosen aus dem Baumarkt arbeiten mit 433 MHz. Die Fernbedienung schickt einen 
abgestimmten Code zum Ein- und Ausschalten an die Funksteckdose. 
Mit dem 433 MHz Empfänger können wir diese Codes ermitteln.
Es gibt mehrere Codevarianten:

⦁ Decimal
⦁ Binary
⦁ Tri-State
⦁ (Stellung der Schiebner eines Dipswitches)

Weiterhin sind zu ermitteln:

⦁ Pulselength
⦁ Protocol

Sind diese Daten bekannt kann man mit dem 433 MHz Sender, anstelle der Fernbedienung, die 
Funksteckdosen schalten

Aufbau

Piezzo Schallwandler an Pin D8
433 MHz Sender an Pin D10

Ermittelte RC-codes
In meinem Fall konnte ich für jedm der 4 Schalter "4 Codes" ermitteln, die in gleicher 
Reihenfolge, sich periodisch wiederholend an den Empfänger geschickt wurden.
Warum?
Dezimal Codes:
A Ein ... A Aus ...
B Ein ... B Aus ...
C Ein ... C Aus ...
D Ein ... D Aus ...

Die Webseite basierend auf "A16_P16_WebserverPiano.ino" dort "index.html" anpassen.

Wie das geht, habe ich beschrieben in "NanoESP Teil 3.pdf"
----------------------------------------------------------------------
<HTML>
<HEAD>
<link rel="shortcut icon" href="data:image/x-icon;," type="image/x-icon">
<title>
NanoESP Funkschalter von Enno
</title>
</HEAD>
<BODY bgcolor="#FFFF99" text="#000000">
<h1>NanoESP Funkschalter</h1>
<hr>
Klicke auf Schalter A, B, C, D "An/Aus"

<br><br>
                 <button onclick="return sendForm(11);"><font size="+1">A An</font></button>
                 <button onclick="return sendForm(10);"><font size="+1">A Aus </font></button><br>
                 <button onclick="return sendForm(21);"><font size="+1">B An </font></button>
                 <button onclick="return sendForm(20);"><font size="+1">B Aus</font> </button><br>
                 <button onclick="return sendForm(31);"><font size="+1">C An </font></button>
                 <button onclick="return sendForm(30);"><font size="+1">C Aus </font></button><br>
                 <button onclick="return sendForm(41);"><font size="+1">D An </font></button>
                 <button onclick="return sendForm(40);"><font size="+1">D Aus </font></button>
<br>
<hr><br>
<script language="JavaScript">
function sendForm(value) {
  var xhr = new XMLHttpRequest();
  xhr.open('POST', "/dataonly", true);

  xhr.send("switch="+value);
  return false; // Prevent page from submitting.
}
</script>

</body>
</HTML>

----------------------------------------------------------------------

Webseite basierend auf  "A16_P16_WebserverPiano.ino" anpassen
Sketch öffnen: > Datei > Sketchbook > NanoESP_WebFunk_02
----------------------------------------------------------------------
/*
  NanoESP_WebFunk_02.ino
  Anpassung 15.07.2018 Enno
  Funkdosen von Enno
  TCP-Server Web-Funksteckdose
  Ändere die SSID und PASSWORD.
*/

#include <RCSwitch.h>
#include <NanoESP.h>
#include <NanoESP_HTTP.h>   //spezielle Library für HTTP-Verbindung
#include <SoftwareSerial.h>
#include "Websites.h"       //Webseite in einer Datei als "progmem"

#define SSID "FRITZ!ENNO"
#define PASSWORD "wlanenno"

#define LED_WLAN 13

NanoESP nanoesp = NanoESP();
NanoESP_HTTP http = NanoESP_HTTP(nanoesp);
RCSwitch mySwitch = RCSwitch();

void setup() {

  Serial.begin(19200);
  nanoesp.init();

  //Nur wenn man die ACCESPOINT Konfiguration ändern möchte
  //nanoesp.configWifi(ACCESSPOINT, "NanoESP", "");
  
  //Einstellungen zum Zugang zum bisher vorhandenen Wlan
  nanoesp.configWifi(DUAL, SSID, PASSWORD);
    
  //Dual Mode
  //nanoesp.configWifiMode(DUAL);
  //delay(1000);
  
  // Warten auf WiFi
  if (nanoesp.wifiConnected())digitalWrite(LED_WLAN, HIGH); 
  else Serial.println(F("Error WLAN Connect"));

  //TCP Server starten
  if (nanoesp.startTcpServer(80)) Serial.println(F("TCP Server Activ")); 
  else Serial.println(F("TCP Server Error"));

  //Anzeige der IP im Monitor
  Serial.println(nanoesp.getIp());

  // Sender ist verbunden mit NanoESP Pin #10  
  mySwitch.enableTransmit(10);
  
  // Protokoll setzen (häufig 1)
  mySwitch.setProtocol(5);

  // Pulslänge setzen
  mySwitch.setPulseLength(500);
}

void loop() {
  String method, ressource, parameter;
  int id, intSwitch;

  if (http.recvRequest(id, method, ressource, parameter)) { //Incoming request, parameters by reference

    Serial.println("New Request from id :" + String(id) + ", method: " + method +  ", ressource: " + ressource +  " parameter: " + parameter);
    
    if (ressource == "/") {
      http.sendFromFlash(id, index, sizeof(index));
      nanoesp.closeConnection(id);
    }

    else if (ressource == "/dataonly") {  //only data is recieved on this url, no need to send a website
      nanoesp.closeConnection(id);
    }

    else {
      http.sendFromFlash(id, error404, sizeof(error404));
      nanoesp.closeConnection(id);
    }

    if (parameter != "")      //the request included parameters
    {
      String strSwitch;
      if (method == GET)  strSwitch = parameter.substring(8); 
      else strSwitch = parameter.substring(7); //get note values
      
      strSwitch = strSwitch.substring(strSwitch.length()-2);
      intSwitch=strSwitch.toInt();
      Serial.println(intSwitch);
      noTone(8);
      tone(8, intSwitch+100, 200);

     switch (intSwitch) {
        case 11:

          mySwitch.send(5279344, 24);
          Serial.println("A Ein");
          break;
        case 10:
          mySwitch.send(5938576, 24);
          Serial.println("A Aus"); 
          break;
        case 21:
          mySwitch.send(6225972, 24);
          Serial.println("B Ein");
          break;
        case 20:
          mySwitch.send(5938576, 24);
          Serial.println("B Aus"); 
          break;
         case 31:
          mySwitch.send(5470028, 24);
          Serial.println("C Ein");
          break;
        case 30:
          mySwitch.send(5794844, 24);
          Serial.println("C Aus"); 
          break;
        case 41:
          mySwitch.send(5429794, 24);
          Serial.println("D Ein");
          break;
        case 40:
          mySwitch.send(5938576, 24);
          Serial.println("D Aus"); 
          break;
      }    
    }
  }
}

----------------------------------------------------------------------
Sketch Hochladen mit Strg+U.
IDE-Monitor öffnen mit Strg+Shift+M.
Unten im Monitor Baudrate auf 19200 Bd einstellen.

Zwischenergebnis
Mit dem NanoESP verfügen wir über die Hardware und einen Webserver.
Ich habe das Projekt "A16_P16_WebserPiano" als Basis genutzt und angepaßt (siehe oben).
Mit dem NanoESP & dem 433 MHz Sender kann ich Funksteckdosen schalten.

Nunmehr steht die Realisierung mit dem Arduino Uno & ESP8266 an.

