
Internet of Things mit dem Arduino & ESP8266 & 433 MHz Funkmodul

Teil 1: Arduino & ESP8266 & 433 MHz Funkmodul

Die Idee:
⦁ Der Arduino hat ein WLAN-Modul erhalten, um ihn aus dem Heimnetz erreichbar zu 

machen. 
⦁ Es sollen handelsübliche Funksteckdosen (433 MHz) gesteuert werden.
⦁ Der Arduino soll sich mit den Funksteckdosen über einen 433 MHz-Sender verbinden.
⦁ Auf dem Arduino soll ein Webserver installiert werden, dessen Homepage z.B. von 

einem Smartphone erreichbar ist. Über das Smartphone sollen die Funksteckdosen 
geschaltet werden.

Die Vorversuche mit dem NanoESP haben die Machbarkeit gezeigt. Nun soll mit dem Arduino 
UNO & einem ESP8266 das Projekt realisiert werden.

Die Teile:

Arduino UNO

Zukaufteile des IoT Projektes

Auflistung der Teile in "Arduino & ESP8266 Teil 0.dbf"

Der Aufbau:

Schematischer Aufbau des NanoESP-Boards (Pretzel-Board / IoT Wifi Board)
In einem ersten Versuch lehne ich mich an diese Schaltung an.

Allerdings benutze ich einen Pegelwandler anstelle des Spannungsteilers.
Die Spannungsversorgung erfolgt über ein USB Breakout Modul und einem 
Spannungswandler.

2



Verschaltung des WLAN-Modules mit dem Controller (Bild Urfassung von Fabian Kainka)
Pinbelegung
Die Einbindung des ESP8266 erfolgt über SoftwareSerial Objekt(receivePin, transmitPin)
Hier: receivePin = D11 (RX) verbunden mit ESP8266 TXD (Pegelwandler)

transmitPin =D 12 (TX) verbunden mit ESP8266 RXD (Pegelwandler)
Sender an D10.

Die Spannungsversorgung
Die 3,3 V des Arduino reichen nicht (Stromstärke) für den ESP8266 aus. Daher eine separate 
Spannungsversorgung von 3,3 V für den ESP8266.

Pegelwandler für Arduino TX -> ESP8266 und für ESP8266 RXD -> an Arduino
Für den Schaltungsaufbau eines Arduino-Uno mit zugekauftem ESP8266 ist zu beachten, das 
die Ausgangsspannung an D12 5 V beträgt, was (s. oben) mit Hilfe einer 
Spannungsteilerschaltung auf 3,3 V reduziert wurde. In Gegenrichtung ist eine Pegelwandlung 
nicht unbedingt erforderlich. Ich nehme die Anpassung mit einem Pegelwandler vor.

Baudrate ARDUINO -> ESP8266
Mein ESP8266-Modul kam mit voreingestellten 115200 Bd, was nach meiner Erfahrung für 
den Arduino UNO ohnehin ein Problem darstellt.
Daher habe ich mit einem "USB ESP8266 ESP01 Seriell Adapter Programmierer"
die Baudrate eingestellt auf 19200 Bd.

Der AT-Befehl lautet= AT+UART_DEF=19200,8,1,0,0
Wichtig: Niemals "AT+IPR=19200" anwenden! Danach ist die serielle Kommunikation nicht 
mehr möglich. Ein neues Aufspielen der Firmware wird nötig.

Der NanoESP hat 19200 Bd eingestellt. Auf diesem Wert beruhen die Libraries von F.Kainka, so 
dass wir den Wert ohnehin auf 19200 Baud einstellen müssen.

ARDUINO & ESP8266 WiFi-Test
Dieser Sketch kommt ohne die NanoESP-Library aus, weil er auf der SoftwareSerial-Library
basiert, die in der Arduino-IDE vorhanden ist. Mit Hilfe von diesem Sketch können AT-Befehle 
im Monitor eingegeben und die Antwort eingesehen werden. 
Sketch über die Zwischenablage in eine neuen Sketch kopieren.
----------------------------------------------------------------------
/*
  SoftwareSerial Beispiel
  Script funktioniert ohne NanoESP-Library
*/

#include <SoftwareSerial.h>     // Ist standardmäßig vorhanden

SoftwareSerial ArduinoESP(11, 12); // Ein Objekt erstellen: NanoESP

void setup()
{
  // Öffne die serielle Kommunikation und warte auf die Verbindung
  Serial.begin(19200);
  // Setze die Datenrate für den "SoftwareSerial port", das ESP8266-Modul
  ArduinoESP.begin(19200);
  
  //Sende einen ersten AT-Befehl um die Antwort des ESP8266-Module zu erkennen
  ArduinoESP.println("AT");
}

void loop()
{
  if (ArduinoESP.available())
    Serial.write(ArduinoESP.read());
  if (Serial.available())
    ArduinoESP.write(Serial.read());
}

----------------------------------------------------------------------
Sketch SoftwareSerial.ino:
In Zeile 5 wird mit "#include" die Library "SoftwareSerial.h" eingebunden.
In Zeile 7 wird das Objekt ArduinoEsp der Klasse "SoftwareSerial" erzeugt.

Mit der Angabe der Pins von RXD und TXD, hier D11 und D12, wird die physikalische
Verbindung  Arduino<->ESP8266 definiert.
In Zeile 12 wird die seriellen Verbindung zum Arduino-Monitor "Serial.begin..." geöffnet.
In Zeile 14 muss auch die serielle Verbindung zum ESP8266 als Objekt "ArduinoEsp " mit der 
Methode "ArduinoEsp .begin..." geöffnet werden. Wichtig hier die voreingestellte Baudrate 
zum ESP8266 mit 19200 Bd, die man nicht verändern darf bzw. kann.
In der "loop" werden in Zeile 23 die Antworten vom ESP8266 an den Monitor geleitet.
In Zeile 25 werden die Eingaben vom Monitor an den ESP8266 gesendet.

Sketch Hochladen mit Strg+U.
IDE-Monitor öffnen mit Strg+Shift+M.
Unten im Monitor Baudrate auf 19200 Bd einstellen (ein Muss).
Weiterhin im Monitor unten "Sowohl NL als auch CR" einstellen (ein Muss für AT-Befehle).
Erwartete Ausgabe (kann abweichend sein)

Einige AT-Befehle: https://iot.fkainka.de/at-befehlsliste
Beachten: Befehle stets in Großbuchstaben.
AT (ohne alles) 
AT+GMR
AT+CWMODE=3
AT+CWLAP
AT+CIFSR
AT+CWJAP=“<ssid>“,“<pass>“

ARDUINO & ESP8266 Test: Webserver & Schaltflächen & Funkmodul

Libraries von F. Kainka "NanoESP-master.zip"
Da ich die Library von F.Kainka nutzen möchte, muss diese installiert sein.
Aktuelle NanoESP Library von F.Kainka:
Browser-Adresse: https://github.com/FKainka/NanoESP
Library nach z.B. "C:\Arduino-18.5\libraries" entpacken.
Hier erscheint die neue Libary: NanoESP-master
Nenne den Ordner um in: NanoESP

RC-Switch Library
Library zum steuern des 433 MHz Senders.
Browser-Adresse: https://github.com/sui77/rc-switch
Library nach z.B. "C:\Arduino-18.5\libraries" entpacken.
Hier erscheint die neue Libary: rc-switch-master
Den Ordner umbenennen in: rc-switch

Sketch basierend auf  "A16_P16_WebserverPiano.ino" anpassen.
Die Webseite habe ich auch nach dem in "NanoESP Teil 3.rtf" beschriebenen Weg angepaßt.
Sketch & Webseite Hochladen mit Strg+U.
IDE-Monitor öffnen mit Strg+Shift+M.
Unten im Monitor Baudrate auf 19200 Bd einstellen.
----------------------------------------------------------------------
/*
  ArduinoESP_WebFunk_02.ino
  Anpassung 15.07.2018 Enno
  Funkdosen von Enno
  TCP-Server Web-Funksteckdose
  Ändere die SSID und PASSWORD.
*/

#include <RCSwitch.h>
#include <NanoESP.h>
#include <NanoESP_HTTP.h>   //spezielle Library für HTTP-Verbindung
#include <SoftwareSerial.h>
#include "Websites.h"       //Webseite in einer Datei als "progmem"

#define SSID "FRITZ!ENNO"
#define PASSWORD "wlanenno"

#define LED_WLAN 13

NanoESP arduinoesp = NanoESP();
NanoESP_HTTP http = NanoESP_HTTP(arduinoesp);
RCSwitch mySwitch = RCSwitch();

void setup() {
  Serial.begin(19200);
  arduinoesp.init();

  //Nur wenn man die ACCESPOINT Konfiguration ändern möchte
  //arduinoesp.configWifi(ACCESSPOINT, "NanoESP", "");
  
  //Einstellungen zum Zugang zum bisher vorhandenen Wlan
  arduinoesp.configWifi(DUAL, SSID, PASSWORD);

  //Dual Mode
  //arduinoesp.configWifiMode(DUAL);
  //delay(1000);
  
  // Warten auf WiFi
  if (arduinoesp.wifiConnected())digitalWrite(LED_WLAN, HIGH); 
  else Serial.println(F("Error WLAN Connect"));

  //TCP Server starten
  if (arduinoesp.startTcpServer(80)) Serial.println(F("TCP Server Activ")); 
  else Serial.println(F("TCP Server Error"));

  //Anzeige der IP im Monitor
  Serial.println(arduinoesp.getIp());

  // Sender ist verbunden mit NanoESP Pin #10  
  mySwitch.enableTransmit(10);
  
  // Protokoll setzen (häufig 1)
  mySwitch.setProtocol(5);

  // Pulslänge setzen
  mySwitch.setPulseLength(500);
}

void loop() {
  String method, ressource, parameter;
  int id, intSwitch;

  if (http.recvRequest(id, method, ressource, parameter)) { //Incoming request, parameters by reference

    Serial.println("New Request from id :" + String(id) + ", method: " + method +  ", ressource: " + ressource +  " 
parameter: " + parameter);
    
    if (ressource == "/") {
      http.sendFromFlash(id, index, sizeof(index));
      arduinoesp.closeConnection(id);
    }

    else if (ressource == "/dataonly") {  //only data is recieved on this url, no need to send a website
      arduinoesp.closeConnection(id);
    }

    else {
      http.sendFromFlash(id, error404, sizeof(error404));
      arduinoesp.closeConnection(id);
    }

    if (parameter != "")      //the request included parameters

    {
      String strSwitch;
      if (method == GET)  strSwitch = parameter.substring(8); 
      else strSwitch = parameter.substring(7); //get note values
      
      strSwitch = strSwitch.substring(strSwitch.length()-2);
      intSwitch=strSwitch.toInt();
      Serial.println(intSwitch);
      noTone(8);
      tone(8, intSwitch+100, 200);

     switch (intSwitch) {
        case 11:
          mySwitch.send(5279344, 24);
          Serial.println("A Ein");
          break;
        case 10:
          mySwitch.send(5794832, 24);
          Serial.println("A Aus"); 
          break;
        case 21:
          mySwitch.send(5980084, 24);
          Serial.println("B Ein");
          break;
        case 20:
          mySwitch.send(5429796, 24);
          Serial.println("B Aus"); 
          break;
         case 31:
          mySwitch.send(5980092, 24);
          Serial.println("C Ein");
          break;
        case 30:
          mySwitch.send(5429804, 24);
          Serial.println("C Aus"); 
          break;
        case 41:
          mySwitch.send(5938578, 24);
          Serial.println("D Ein");
          break;
        case 40:
          mySwitch.send(5470018, 24);
          Serial.println("D Aus"); 
          break;
      }    
    }
  }
}

----------------------------------------------------------------------

Erwartete Ausgabe im IDE-Monitor:

Webseite im PC-Browser mit der IP 192.168.178.33" aufrufen.
Erwartete Ansicht:

Duch Klicken auf die Schaltflächen lassen sich die Funksteckdosen schalten.

HTML-Code der Webseite:
----------------------------------------------------------------------
<HTML>
<HEAD>
<link rel="shortcut icon" href="data:image/x-icon;," type="image/x-icon">
<title>
Arduino UNO & ESP8266 & Funkschalter
</title>
</HEAD>
<BODY bgcolor="#FFFF99" text="#000000">
<h1>Arduino UNO & ESP8266 & Funkschalter</h1>
<hr>
Klicke auf Schalter A, B, C, D "An/Aus"

<br><br>

                 <button onclick="return sendForm(11);"><font size="+1">A An</font></button>
                 <button onclick="return sendForm(10);"><font size="+1">A Aus </font></button><br>
                 <button onclick="return sendForm(21);"><font size="+1">B An </font></button>
                 <button onclick="return sendForm(20);"><font size="+1">B Aus</font> </button><br>
                 <button onclick="return sendForm(31);"><font size="+1">C An </font></button>
                 <button onclick="return sendForm(30);"><font size="+1">C Aus </font></button><br>
                 <button onclick="return sendForm(41);"><font size="+1">D An </font></button>
                 <button onclick="return sendForm(40);"><font size="+1">D Aus </font></button>
<br>
<hr><br>
<script language="JavaScript">
function sendForm(value) {
  var xhr = new XMLHttpRequest();
  xhr.open('POST', "/dataonly", true);

  xhr.send("switch="+value);
  return false; // Prevent page from submitting.
}
</script>

</body>
</HTML>

----------------------------------------------------------------------

Ergebnis
Mit dem Arduino & ESP8266 verfügen wir über die Hardware und einen Webserver.
Mit dem Arduino & ESP8266 & dem 433 MHz Sender kann man Funksteckdosen schalten.

10


